Denoising using wavelets

This Mathematica notebook uses a package for discrete wavelet transforms
by P. van Fleet with examples taken from his book *Discrete Wavelet Transforms*

In[2]:=
```
<< DiscreteWavelets`DiscreteWavelets`
```

SetDelayed::write : Tag Entropy in Entropy[v_] is Protected. >>

Signal and noise

In[3]:=
```
(*the Heavisine function*)

f[t_] := 4 Sin[4 Pi t] - Sign[t - 0.3] - Sign[.72 - t]
```

In[4]:=
```
Plot[f[t], {t, 0, 1}]
```

Out[4]=



In[5]:=
```
(* Discretized version of the Heavisine
 function with 2048 evaluation points in the interval [0,1] *)

v = Table[f[k / 2048], {k, 0, 2047}];
p1 = ListPlot[v, Joined → True, PlotLabel → "Heavisine function"];

(* Adding white noise with noise level σ = 0.25 *)

n = Length[v];
nd = NormalDistribution[0., 1.];
SeedRandom[];
noise = Table[Random[nd], {k, 1, n}];
sigma = .25;
w = v + sigma * noise;
p2 = ListPlot[w, Joined → True, PlotLabel → "noisy Heavisine"];
GraphicsGrid[{{p1}, {p2}}, Frame → All, ImageSize → Full]
```

Out[14]=



In[15]:=

```
Norm[v - w]
```

Out[15]=

```
11.1736
```

Denoising using 4-level D-12 transform

In[16]:=
```
its = 4;
wtd12 = WT1D[w, Daub[12], NumIterations → its];
wtd12list = WaveletVectorToList[wtd12, NumIterations → its];
hp1d12 = Last[wtd12list];
```

In[20]:=
```
(* estimating the noise level from the first
 detail component of the transformed signal (using D12)*)
σ = MAD[hp1d12] / .6745;
Print["estimate for the noise level is ", σ, "."];
```

```
estimate for the noise level is 0.251154.
```

In[22]:=
```
(* universal tolerance after Donoho-Johnstone *)

n = Length[wtd12] - Length[First[wtd12list]];
λ = sigma * Sqrt[2 * Log[ n]];
Print["the universal tolerance is λ = ", λ, "."];
```

```
the universal tolerance is λ = 0.972116.
```

In[25]:=
```
(* applying the threshold operation with universal tolerance*)

lpd12 = First[wtd12list];
hpd12 = Flatten[Drop[wtd12list, 1]];
newhpd12 = Map[ShrinkageFunction[#, λ] &, hpd12];
newwtd12 = Join[lpd12, newhpd12];
```

In[29]:=
```
WaveletVectorPlot[wtd12,
 NumIterations → its,
 PlotRange → All,
 PlotLabel -> "Wavelet transform of the noised signal", UseColors -> True]
```

Out[29]=



In[30]:=
```
WaveletVectorPlot[newwtd12,
 NumIterations → its,
 PlotLabel -> "threshold operation",
 UseColors -> True]
```

Out[30]=



In[31]:=
```
WaveletVectorPlot[hpd12, NumIterations → its, PlotRange → All,
 PlotLabel -> "WT of the noised signal -- detailed view", UseColors -> True]
```

Out[31]=



In[32]:=
```
WaveletVectorPlot[newhpd12,
 NumIterations → its,
 PlotLabel -> "threshold operation -- detailed view", UseColors -> True]
```

Out[32]=

In[33]:=
```
denoisedd12 = IWT1D[newwtd12,
   Daub[12],
   NumIterations → its];
ListPlot[denoisedd12,
 Joined → True,
 PlotLabel -> "Denoised signal"]
```

Out[34]=



In[35]:=
```
ListPlot[{v, denoisedd12},
 Joined → True,
 PlotStyle → {Black, Red},
 PlotLabel -> "Denoised signal, compared to the original"]
```

Out[35]=



In[36]:=
```
Norm[v - denoisedd12]
```

Out[36]=
```
4.21469
```

Denoising using 6-level Coiflet-12 transform

In[37]:=
```
its = 6;

univc12 = UniversalThreshold[w,
    N[Coif[2]],
    NumIterations → its];
denoisedc12 = WaveletShrinkage[w,
    N[Coif[2]],
    univc12,
    NumIterations → its];

ListPlot[{denoisedc12},
 Joined → True,
 PlotLabel -> "Denoised signal"]

ListPlot[{v, denoisedc12},
 Joined → True,
 PlotStyle → {Black, Red},
 PlotLabel -> "Denoised signal, compared to original"]
```

Out[40]=



Out[41]=



In[42]:=
```
Norm[v - denoisedc12]
```

Out[42]=
```
4.10651
```

Denoising using 6-level Haar transform

In[43]:=
```
its = 6;

univh = UniversalThreshold[w, N[Haar[]], NumIterations → its];
denoisedh = WaveletShrinkage[w, N[Haar[]], univh, NumIterations → its];

ListPlot[{v, denoisedh},
 Joined → True,
 PlotStyle → {Black, Red},
 PlotLabel -> "denoised signal, compared to the original"]
```

Out[46]=



In[47]:=
```
Norm[v - denoisedh]
```

Out[47]=
```
8.71634
```